Nominal Isabelle^1

Christian Urban Stefan Berghofer with contributions from Julien Narboux December 22, 2008

1This manual is released while being written. The hope is that it is still useful to others.

Contents

1 Introduction 3 1.1 Installation . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4

2 A Quick Example 6

3 Nominal Reasoning Infrastructure 19 3.1 Preliminaries . . . . . . . . . . . . . . . . . . . . . . . . . . . 19 3.2 Atom Declarations . . . . . . . . . . . . . . . . . . . . . . . . 20 3.3 Permutations . . . . . . . . . . . . . . . . . . . . . . . . . . . 21 3.4 Support . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 24 3.5 Freshness . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 26 3.6 Permutation Types and Finitely Supported Types . . . . . . . . 27 3.7 Abstractions and Alpha-Equivalence . . . . . . . . . . . . . . 29

4 Nominal Datatypes 30 4.1 Declaration . . . . . . . . . . . . . . . . . . . . . . . . . . . . 30 4.2 Strong Structural Inductions . . . . . . . . . . . . . . . . . . . 33 4.3 Equivariance Lemmas . . . . . . . . . . . . . . . . . . . . . . 36 4.4 Function Definitions . . . . . . . . . . . . . . . . . . . . . . . 37 4.5 Inductive Definitions and Strong Rule Induction Principles . . 38

5 Advanced Topics 46 5.1 Functions That Need to Generate Fresh Names . . . . . . . . 46

6 Examples 48

A Frequently Asked Questions 50 A.1 The atom decl command does not work. . . . . . . . . . . . . 50 A.2 Can one avoid typing \<guillemotleft>? . . . . . . . . . . . 50 A.3 I want to prove something about support or freshness, but theorems which should be trivial cannot be proved. . . . . . . 50

B Infrastructure 52 B.1 Equivariance Lemmas . . . . . . . . . . . . . . . . . . . . . . 52

Chapter 1

Introduction

Dealing with binders, renaming of bound variables, capture-avoiding substitution, etc., is very often a major problem in formal proofs, especially in proofs by structural and rule induction. Nominal Isabelle is designed to make such proofs easy to formalise: it provides an infrastructure for declaring nominal datatypes (that is alpha-equivalence classes) and for defining functions over them by structural recursion. It also provides induction principles that have Barendregt’s variable convention already built in. At present Nominal Isabelle is still under heavy development, and also theoretical and implementation issues still remain unsolved. Nevertheless it can and has already been used to formalise a number of results about languages involving binders. This includes the standard strong normalisation proof for the simply-typed lambda-calculus, the Tait/Martin-Löf proof and Takahshi/Pollack proof for the Church-Rosser property of beta-reduction, typical proofs from structural operational semantics, a chapter by Crary on logical relations, a paper by Harper and Pfenning on LF and parts of the PoplMark Challenge. Urban also used it to verify the results in his PhDthesis on strong normalisation of cut-elimination in classical logic.

There is a mailing list for Nominal Isabelle to which users are very welcome to subsribe. Details can be found under “Mailing List” at

http://isabelle.in.tum.de/nominal

A great source of information about Isabelle and its old-style proof-scripts is the tutorial by Nipkow et al. [2002]. It can be downloaded in the “Documentation” section at

http://isabelle.in.tum.de/documentation.html

In the same place is a short tutorial about the Isar proof-language developed by Wenzel. This manual assumes that the reader is familiar with Isabelle (more precisely Isabelle/HOL) or learns it by using these notes as a source of examples and by consulting the material referred above for further information.

1.1 Installation

To run Nominal Isabelle you need four packages: PolyML, Isabelle, an emacs and Proof-General. Nominal Isabelle is part of the Isabelle distribution, which you can find at

http://isabelle.in.tum.de

We make occasionally “nominal” releases, which can be downloaded from the “Download” section at

http://isabelle.in.tum.de/nominal

To unpack and install Nominal Isabelle and the additional packages (such as Proof-General and PolyML) follow the installation notes of Isabelle. You can find them in the file

[ISABELLE HOME]/INSTALL

where [ISABELLE HOME] refers to the directory in which you have unpacked the Isabelle sources. The build process for Nominal Isabelle needs to be started inside this directory with the command:

./build -m HOL-Nominal

There are no pre-compiled heap files for Nominal Isabelle. After the build completes (this can take a few minutes), the files need to be installed with the command

./bin/isabelle install -p name-of-a-directory

where the directory is a place in which the system can find executable files. Under Linux this is typically /usr/local/bin.

Isabelle together with the Proof-General interface can be started with isabelle emacs name-of-a-theory-file &

You might have to explicitly specify where Isabelle can find your emacs by providing an option -p path-to-emacs.

This way of starting Isabelle means that the nominal theory will be rechecked when you start your theory. This results in a delay of approximately 30 seconds at the beginning of your interactive session. The delay can be avoided by enabling the option “HOL-Nominal” in the menu “Isabelle → Logics”. Unfortunately, one cannot save the state of this option. So if you always want to have HOL-Nominal turned on, then you can either start Isabelle with the command

isabelle emacs -l HOL-Nominal name-of-a-theory-file &

or set the environment variable

ISABELLE_LOGIC="HOL-Nominal"

in your seetings file for Isabelle (usually under ∼/.isabelle/etc).

Also when Isabelle is first started, check that the X-Symbols are enabled (see “Proof-General → Options”) and save any changes (see “Proof-General → Options → Save Options”). The package X-Symbol is not essential for working with Isabelle, but can greatly enhance the readability of proof scripts.

Nominal Isabelle contains numerous examples of formalisations for properties about terms with binders. See Chapter 6.

Chapter 2

A Quick Example

This chapter provides a quick overview over Nominal Isabelle by giving the complete proof of the Church-Rosser theorem. The proof we show is due to Takahashi [1995] with some improvements by Pollack [1995].

Informal reasoning about lambda-terms often starts by stating that there is a countably infinite supply of variables. The formal equivalent in Nominal Isabelle for this statement is the declaration

atom decl name

This declaration introduces a type name whose elements can stand for binders. They are needed when we want to define lambda-terms using the command nominal datatype.

nominal datatype lam = Var ”name” | App ”lam” ”lam” | Lam ”≪name≫lam” (”Lam [ ]. ”)

The ≪name≫lam indicates that a name is bound in the term that follows. By convention we introduce the syntax annotation [ ]. for binders. In this way we can write Lam [x].t, instead of Lam x t. Other conventions are possible. Unfortunately, however, the usual syntax λx. t is already used by Isabelle’s HOL-logic.

The main point of using Nominal Isabelle is that the definition above results in alpha-equated lambda-terms. We can, for example, show the following equality:

lemma shows ”Lam [x].Var x = Lam [y].Var y” by (auto simp add: lam.inject alpha swap simps fresh atm)

This means the HOL-logic allows us without any further ado to replace in any context a lambda-term by an alpha-equivalent lambda-term. A lot of convenience in Nominal Isabelle arises from this fact.

The most important operation we need for lambda-terms is capture-avoiding substitution. This operation can be defined in Nominal Isabelle as total function using the command nominal primrec:

nominal primrec subst :: ”lam ⇒ name ⇒ lam ⇒ lam” (” [ ::= ]”) where ”(Var x)[y::=s] = (if x=y then s else (Var x))” | ”(App t_1 t_2)[y::=s] = App (t_1[y::=s]) (t_2[y::=s])” | ”x ♯ (y,s) =⇒ (Lam [x].t)[y::=s] = Lam [x].(t[y::=s])”

By convention we use the notation t[y::=s] for writing substitutions (the more obvious convention t[y:=s] is unfortunately already used in Isabelle for updating maps).

The crucial line in the definition above is the third one, which includes the precondition x ♯ (y, s) standing for the variable x being fresh for the variable y and the lambda-term s. Clearly, only in this case we should be able to move a substitution under the binder, otherwise we defined a function that does not respect alpha-equivalence. It is actually easy to write down functions over lambda-terms that do not respect alpha-equivalence and thus it is easy to prove faulty statements. Nominal Isabelle prevents this by requiring us to prove several properties about the definition of substitution. The corresponding subgoals can be discharged by the following five lines

apply(finite guess)+ apply(rule TrueI)+ apply(simp add: abs fresh) apply(fresh guess)+ done

which are typical for many function definitions over lambda-terms.

The notion ♯ about freshness is central to Nominal Isabelle and automatically defined for nominal datatypes and common datatypes, such as products, lists, natural numbers and so on. A variable being fresh is in most cases equivalent to the variable not being free.

An important lemma we need to prove about substitution is the substitution lemma. To establish it, we need two facts about freshness and substitution. The first states that substitutions can be “forgotten” provided the variable that is being substituted is not free (i.e. is fresh).

lemma forget: assumes a: ”x ♯ t” shows ”t[x::=s] = t” using a by (nominal induct t avoiding: x s rule: lam.strong induct) (auto simp add: abs fresh fresh atm)

The second is about conditions when a variable is fresh for a substitution: more precisely it states that the variable z will be fresh for z ♯ t[y::=s] provided that z is fresh for s and either z = y or z also fresh for t. This can be proved as follows:

lemma fresh fact: fixes z::”name” assumes a: ”z ♯ s” ”z=y ∨ z ♯ t” shows ”z ♯ t[y::=s]” using a by (nominal induct t avoiding: z y s rule: lam.strong induct) (auto simp add: abs fresh fresh prod fresh atm)

With these two facts at our disposal, the substitution lemma can then be easily established.

lemma substitution lemma: assumes a: ”x≠y” ”x ♯ u” shows ”t[x::=s][y::=u] = t[y::=u][x::=s[y::=u]]” using a by (nominal induct t avoiding: x y s u rule: lam.strong induct) (auto simp add: fresh fact forget)

It is a fun exercise to unfold the automation in this proof: one ends up with a formal proof that looks uncannily like the informal proof in Barendregt [1981]. If you do not like to do this exercise yourself, have a look in the file

CR.thy.

While the previous three lemmas are quite standard and can be found in many informal Church-Rosser proofs, we need two more facts about substitution. First we establish that substitution is equivariant. As we will see later on, much of the reasoning in Nominal Isabelle depends on this property. Equivariance for substitution states that a permutation (a bijective renaming of variables) applied on the “outside” of a substitution can be pushed inside to the arguments. Permutations and a permutation applied to a term are concepts provided by Nominal Isabelle. Applying a permutation is written infix as^• .

lemma subst eqvt[eqvt]: fixes π::”name prm” shows ”π^•(t[x::=s]) = (π^•t)[(π^•x)::=(π^•s)]” by (nominal induct t avoiding: x s rule: lam.strong induct) (auto simp add: perm bij fresh atm fresh bij)

Since this property is needed frequently “behind the scenes”, we make Nominal Isabelle aware of it by giving this lemma the attribute [eqvt]. In two of the later lemmas it will also be necessary to rename a variable being substituted with a fresh variable. This is formalised in the following lemma:

lemma subst rename: assumes a: ”y ♯ t” shows ”t[x::=s] = ([(y,x)]^•t)[y::=s]” using a by (nominal induct t avoiding: x y s rule: lam.strong induct) (auto simp add: swap simps fresh atm abs fresh)

where [(y, x)]^• t stands for the swapping of y and x in t. A swapping is a permutation consisting of a single pair of variables. You can think of it as the simultaneous renaming of y for x and x for y.

Next we need the definition for beta-reduction. It can be defined as inductive predicate using the Isabelle command inductive.

inductive ”Beta” :: ”lam ⇒ lam ⇒ bool” (” −→_β ”) where b1[intro]: ”t_1 −→_β t_2 =⇒ App t_1 s −→_β App t_2 s” | b2[intro]: ”s_1 −→_β s_2 =⇒ App t s_1 −→_β App t s_2” | b3[intro]: ”t_1 −→_β t_2 =⇒ Lam [x].t_1 −→_β Lam [x].t_2” | b4[intro]: ”App (Lam [x].t) s −→_β t[x::=s]”

We introduce the readable notation −→_β for beta-reduction and we also tag each rule with the attribute [intro] so that the automatic proofsearch tools of Isabelle are aware of these rules. The first three rules in this definition are the usual concruence rules and the fourth is the betacontraction. In Nominal Isabelle it is also possible to define beta-reduction via the notion of a context with a hole (see the examples in Context.thy and CK Machine.thy).

The transitive closure of beta-reduction is again an inductive definition.

inductive ”Beta star” :: ”lam ⇒ lam ⇒ bool” (” −→_β^∗ ”) where bs1[intro]: ”t −→_β^∗ t” | bs2[intro]: ”t −→_β s =⇒ t −→_β^∗ s” | bs3[intro]: ”[[t_1−→_β^∗ t_2; t_2 −→_β^∗ t_3]] =⇒ t_1 −→_β^∗ t_3”

There is an equivalent definition for the transitive closure of −→_β involving only two rules, but the version above will turn out to be more convenient in some of the subsequent proofs.

The point of the Church-Rosser proof is to show that every two −→_β^∗- reductions can be joined. No proof in the literature, however, establish this property directly for this reduction relation. Instead an auxiliary reduction relation, written −→_1, is introduced for which it is easier to establish Church-Rosser. It can also be shown that its transitive closure is equivalent to −→_β^∗. The main reasoning given below will, therefore, be over −→_1 defined next.

inductive One :: ”lam ⇒ lam ⇒ bool” (” −→_1 ”) where o1[intro]: ”Var x−→_1 Var x” | o2[intro]: ”[[t_1−→_1t_2; s_1−→_1s_2]] =⇒ App t_1 s_1 −→_1 App t_2 s_2” | o3[intro]: ”t_1−→_1t_2 =⇒ Lam [x].t_1 −→_1 Lam [x].t_2” | o4[intro]: ”[[x ♯ (s_1,s_2); t_1−→_1t_2; s_1−→_1s_2]] =⇒ App (Lam [x].t_1) s_1 −→_1 t_2[x::=s_2]”

This definition is similar to what one can find in the literature, except for the last rule where the freshness condition x ♯ (s_1, s_2) is included as a premise. This premise is needed in Nominal Isabelle in order to derive a stronger induction principle for this definition. This stronger induction principle has the usual variable convention already built in and so will make the reasoning about this definition much more convenient. In order to derive the stronger induction principle, Nominal Isabelle first has know that this definition is equivariant. Since it knows that substitution is equivariant (remember the [eqvt] attribute on the corresponding lemma) and also that freshness is equivariant, equivariance for −→_1 can be derived automatically by stating:

equivariance One

This gives us the property

If t −→_1 s then π^• t −→_1 π^• s.

named One.eqvt. We need it in one lemma below and as mentioned above also need it in the next nominal inductive statement:

nominal inductive One by (simp all add: abs fresh fresh fact)

This statement derives for us the stronger induction principle for −→_1. To do so, it checks that the binder in the rules o3 and o4 are not free in the conclusions of those rules. This fact is clear for o3 with the conclusion

Lam [x].t_1 −→_1 Lam [x].t_2

but for the conclusion of o4

App (Lam [x].t_1) s_1 −→_1 t_2[x::=s_2]

it only holds if we have the premise x ♯ (s_1, s_2).

If we do not do this test, then deriving the stronger induction principle for an inductive definition can potentially lead to faulty proofs. See the examples in VC Condition.thy where an inconsistency is derived using the variable convention. The unfortunate point with needing this additional premise in rule o4 is that it makes this rule harder to apply; it has no influence on what reductions can be derived, because we can show that this premise is actually superfluous:

lemma better o4 intro: 1 assumes a: ”t_1 −→_1 t_2” ”s_1 −→_1 s_2” 2 shows ”App (Lam [x].t_1) s_1 −→_1 t_2[x::=s_2]” 3 proof - 4 obtain y::”name” where fs: ”y ♯ (x,t_1,s_1,t_2,s_2)” 5 by (rule exists fresh, rule fin supp, blast) 6 have ”App (Lam [x].t_1) s_1 = App (Lam [y].([(y,x)]^•t_1)) s_1” using fs 7 by (auto simp add: lam.inject alpha‘ fresh prod fresh atm) 8 also have ”. . . −→_1 ([(y,x)]^•t_2)[y::=s_2]” using fs a 9 by (auto simp add: One.eqvt) 10 also have ”. . . = t_2[x::=s_2]” using fs by (simp add: subst rename[symmetric]) 11 finally show ”App (Lam [x].t_1) s_1 −→_1 t_2[x::=s_2]” by simp 12 qed 13 The proof follows a quite standard pattern: a fresh name y is chosen in lines 5 and 6. Line 7 alpha-renames the term before the reduction. In line 9 we apply the introduction rule of −→_1 for the alpha-renamed term (this requires the property One.eqvt for the premise t_1 −→_1 t_2). Then we rename the substitution back to the original term in line 11.

To make progress in our Church-Rosser proof, we need to establish two properties for the reduction −→_1: reflexivity and substitutivity.

lemma One refl: shows ”t −→_1 t” by (nominal induct t rule: lam.strong induct) (auto)

lemma One subst:

assumes a: ”t_1 −→_1 t_2” ”s_1 −→_1 s_2” shows ”t_1[x::=s_1] −→_1 t_2[x::=s_2]” using a by (nominal induct t_1 t_2 avoiding: s_1 s_2 x rule: One.strong induct) (auto simp add: substitution lemma fresh atm fresh fact)

In the proof of the substitutivity property it is essential to have the stronger induction principle for −→_1 at our disposal, because otherwise there is no convenient way to move the substitutions under binders. We also need later on the following four inversion principles for −→_1. We derive them as elimination rules using obtains.

lemma One Var:

assumes a: ”Var x −→_1 t” obtains ”t = Var x” using a by (cases rule: One.cases) (simp all)

lemma One Lam:

assumes a: ”Lam [x].t −→_1 s” ”x ♯ s” obtains t‘ where ”s = Lam [x].t‘” ”t −→_1 t‘” using a by (cases rule: One.strong cases)

(auto simp add: lam.inject abs fresh alpha)

lemma One App:

assumes a: ”App t s −→_1 r” obtains t‘ s‘ where ”r = App t‘ s‘” ”t −→_1 t‘” ”s −→_1 s‘” | x p p‘ s‘ where ”r = p‘[x::=s‘]” ”t = Lam [x].p” ”p −→_1 p‘” ”s −→_1 s‘” ”x ♯ (s,s‘)” using a by (cases rule: One.cases)

(auto simp add: lam.inject)

lemma One Redex:

assumes a: ”App (Lam [x].t) s −→_1 r” ”x ♯ (s,r)” obtains t‘ s‘ where ”r = App (Lam [x].t‘) s‘” ”t −→_1 t‘” ”s −→_1 s‘” | t‘ s‘ where ”r = t‘[x::=s‘]” ”t −→_1 t‘” ”s −→_1 s‘” using a by (cases rule: One.strong cases)

(auto elim!: One Lam simp add: lam.inject abs fresh alpha fresh prod)

The transitive closure of −→_1 is defined next.

inductive ”One star” :: ”lam ⇒ lam ⇒ bool” (” −→_1^∗ ”) where os1[intro]: ”t −→_1^∗ t” | os2[intro]: ”t −→_1 s =⇒ t −→_1^∗ s” | os3[intro]: ”[[t_1−→_1^∗ t_2; t_2 −→_1^∗ t_3]] =⇒ t_1 −→_1^∗ t_3”

The beautiful trick in the Church-Rosser proof by Takahashi is to introduce a third reduction relation for which a triangle property can be easily established.

inductive Dev :: ”lam ⇒ lam ⇒ bool” (” −→_d ”) where d1[intro]: ”Var x −→_d Var x” | d2[intro]: ”t −→_d s =⇒ Lam [x].t −→_d Lam[x].s” | d3[intro]: ”[[¬(∃ y t‘. t_1 = Lam [y].t‘); t_1 −→_d t_2; s_1 −→_d s_2]] =⇒ App t_1 s_1 −→_d App t_2 s_2” | d4[intro]: ”[[x ♯ (s_1,s_2); t_1 −→_d t_2; s_1 −→_d s_2]] =⇒ App (Lam [x].t_1) s_1 −→_d t_2[x::=s_2]”

Note that this reduction relation is defined so that no rule overlaps with another. It will be crucial again to use a stronger induction principle with this definition, so we have to show equivariance and “nominal inductiveness”.

equivariance Dev nominal inductive Dev by (simp all add: abs fresh fresh fact)

Like with −→_1 the proof in nominal inductive will only go through if we have the premise x ♯ (s_1, s_2) in rule d4. However, it is otherwise unnecessary and therefore we derive a better introduction rule.

lemma better d4 intro:

assumes a: ”t_1 −→_d t_2” ”s_1 −→_d s_2” shows ”App (Lam [x].t_1) s_1 −→_d t_2[x::=s_2]” proof - obtain y::”name” where fs: ”y ♯ (x,t_1,s_1,t_2,s_2)” by (rule exists fresh, rule fin supp,blast) have ”App (Lam [x].t_1) s_1 = App (Lam [y].([(y,x)]^•t_1)) s_1” using fs by (auto simp add: lam.inject alpha‘ fresh prod fresh atm) also have ”. . . −→_d ([(y,x)]^•t_2)[y::=s_2]” using fs a by (auto simp add: Dev.eqvt) also have ”. . . = t_2[x::=s_2]” using fs by (simp add: subst rename[symmetric]) finally show ”App (Lam [x].t_1) s_1 −→_d t_2[x::=s_2]” by simp qed

We will need an inversion principle for the reduction −→_d and for this it will be advantageous to know that freshness is preserved by the −→_d reduction.

lemma Dev preserves fresh: fixes x::”name” assumes a: ”t−→_d s” shows ”x ♯ t =⇒ x ♯ s” using a by (induct) (auto simp add: abs fresh fresh fact)

The next lemma characterises the term to which a lambda-abstraction dreduces. Since we are going to invert a rule involving a binder, the inversion only works if the binder is fresh with respect to the judgement to be inverted. If we do not impose this restriction, we obtain only a much weaker statement that is inconvenient to use.

To do the inversion, we know in the proof below that x ♯ Lam [x].t holds (Line 5) and so by the preceding lemma also x ♯ s (Line 6). With these two fact at our disposal, we can infer the form of the reduct.

lemma Dev Lam: 1 assumes a: ”Lam [x].t −→_d s” 2 shows ”∃ s‘. s = Lam [x].s‘ ∧ t −→_d s‘” 3 proof - 4 from a have ”x ♯ Lam [x].t” by (simp add: abs fresh) 5 with a have ”x ♯ s” by (simp add: Dev preserves fresh) 6 with a show ”∃ s‘. s = Lam [x].s‘ ∧ t −→_d s‘” 7 by (cases rule: Dev.strong cases) 8 (auto simp add: lam.inject abs fresh alpha) 9 qed 10 We can now show that for every lambda-term there exists a d-reduct.

lemma Development existence: shows ”∃ t‘. t −→_d t‘” by (nominal induct t rule: lam.strong induct) (auto dest!: Dev Lam intro: better d4 intro)

The reason for introducing the reduction −→_d is that we can show a “little” Church-Rosser lemma, namely that a d-reduction and a 1-reduction can always be joined.

lemma Triangle:

assumes a: ”t −→_d t_1” ”t −→_1 t_2” shows ”t_2 −→_1 t_1” using a

We prove this lemma by induction on t −→_d t_1. Any binder that is used in the proof should be fresh for t_2. This requires us to set up the induction as follows:

proof (nominal induct avoiding: t_2 rule: Dev.strong induct)

The only interesting case is d4 where we have to be careful to apply the One Redex inversion instead of One App. It might be possible to make this proof completely automatic on the expense of folding these two inversion principles into one complicated case analysis. However below we are not doing this and instead analyse the case explicitly:

case (d4 x s_1 s_2 t_1 t_1‘ t_2) have fc: ”x ♯ t_2” ”x ♯ s_1” by fact+ have ”App (Lam [x].t_1) s_1 −→_1 t_2” by fact

The fc assumptions are needed in order to invert the assumption about how App (Lam [x].t_1) s_1 reduces. The inversion gives us the following two possibilities: either some term inside the redex reduces, or the redex itself.

then obtain t‘ s‘ where reds:

”(t_2 = App (Lam [x].t‘) s‘ ∧ t_1 −→_1 t‘ ∧ s_1 −→_1 s‘) ∨ (t_2 = t‘[x::=s‘] ∧ t_1 −→_1 t‘ ∧ s_1 −→_1 s‘)” using fc by (auto elim!: One Redex)

Having obtained the terms t‘ and s‘, we can instantiate the two induction hypotheses in this case.

have ih1: ”t_1 −→_1 t‘ =⇒ t‘ −→_1 t_1‘” by fact have ih2: ”s_1 −→_1 s‘ =⇒ s‘ −→_1 s_2” by fact

Now we only have to analyse the two cases and in each of them establish that t_2 reduces to t_1‘[x::=s_2].

{ assume ”t_1 −→_1 t‘” ”s_1 −→_1 s‘” then have ”App (Lam [x].t‘) s‘ −→_1 t_1‘[x::=s_2]” using ih1 ih2 by (auto intro: better o4 intro) } moreover { assume ”t_1 −→_1 t‘” ”s_1 −→_1 s‘” then have ”t‘[x::=s‘] −→_1 t_1‘[x::=s_2]” using ih1 ih2 by (auto intro: One subst) } This allows us to conclude this case.

ultimately show ”t_2 −→_1 t_1‘[x::=s_2]” using reds by auto qed (auto elim!: One App One Var One Lam)

The other three cases can be discharged by the automatic proof-search tools. Using the triangle property and the existence of a development, it is now straightforward to establish the diamond property for −→_1.

lemma Diamond for One: assumes a: ”t −→_1 t_1” ”t −→_1 t_2” shows ”∃ t_3. t_2 −→_1 t_3 ∧ t_1 −→_1 t_3” using a by (metis Development existence Triangle)

We only need to string together the lemma about the existence of a d-reduct and then use the triangle lemma to prove that the reductions are joinable. And from this the following rectangle property can be easily deduced.

lemma Rectangle for One: assumes a: ”t −→_1^∗ t_1” ”t −→_1 t_2” shows ”∃ t_3. t_1 −→_1 t_3 ∧ t_2 −→_1^∗ t_3” using a Diamond for One by (induct arbitrary: t_2) (blast)+

Finally we can show Church-Rosser for −→_1^∗.

lemma CR for One star:

assumes a: ”t −→_1^∗ t_1” ”t −→_1^∗ t_2” shows ”∃ t_3. t_2 −→_1^∗ t_3 ∧ t_1 −→_1^∗ t_3” using a Rectangle for One by (induct arbitrary: t_2) (blast)+

It remains to show that the reductions −→_1^∗ and −→_β^∗ are equivalent. This involves tedious proofs that unfortunately cannot be automated much. First we show the following three congruence rules for −→_β^∗.

lemma Beta Lam cong: assumes a: ”t_1 −→_β^∗ t_2” shows ”Lam [x].t_1 −→_β^∗ Lam [x].t_2” using a by (induct) (blast)+ lemma Beta App cong aux: assumes a: ”t_1 −→_β^∗ t_2” shows ”App t_1 s−→_β^∗ App t_2 s” and ”App s t_1 −→_β^∗ App s t_2” using a by (induct) (blast)+ lemma Beta App cong:

assumes a: ”t_1 −→_β^∗ t_2” ”s_1 −→_β^∗ s_2” shows ”App t_1 s_1 −→_β^∗ App t_2 s_2” using a by (blast intro: Beta App cong aux) lemmas Beta congs = Beta Lam cong Beta App cong

These congruence rules allow us to show that a single 1-reduction can be matched by zero or more beta-reductions.

lemma One implies Beta star: assumes a: ”t −→_1 s” shows ”t −→_β^∗ s” using a by (induct) (auto intro!: Beta congs)

For the other direction we need to show the following congruence lemma for the reduction −→_1^∗.

lemma One congs: assumes a: ”t_1 −→_1^∗ t_2” shows ”Lam [x].t_1 −→_1^∗ Lam [x].t_2” and ”App t_1 s −→_1^∗ App t_2 s” and ”App s t_1 −→_1^∗ App s t_2” using a by (induct) (auto intro: One refl)

A consequence is now that a single beta-reduction can be matched by zero or more 1-reductions.

lemma Beta implies One star: assumes a: ”t_1 −→_β t_2” shows ”t_1 −→_1^∗ t_2” using a by (induct) (auto intro: One refl One congs better o4 intro)

Stringing both directions together gives us the lemma that the reduction −→_1^∗ and −→_β^∗ are in fact equal.

lemma Beta star equals One star: shows ”t_1 −→_1^∗ t_2 = t_1 −→_β^∗ t_2” proof assume ”t_1 −→_1^∗ t_2” then show ”t_1 −→_β^∗ t_2” by (induct) (auto intro: One implies Beta star) next assume ”t_1 −→_β^∗ t_2” then show ”t_1 −→_1^∗ t_2” by (induct) (auto intro: Beta implies One star) qed Since we have already established that −→_1^∗ is Church-Rosser, the reduction −→_β^∗ must be too.

theorem CR for Beta star:

assumes a: ”t −→_β^∗ t_1” ”t −→_β^∗ t_2” shows ”∃ t_3. t_1 −→_β^∗ t_3 ∧ t_2 −→_β^∗ t_3” proof - from a have ”t −→_1^∗ t_1” and ”t−→_1^∗ t_2” by (simp all add: Beta star equals One star) then have ”∃ t_3. t_1 −→_1^∗ t_3 ∧ t_2 −→_1^∗ t_3” by (simp add: CR for One star) then show ”∃ t_3. t_1 −→_β^∗ t_3 ∧ t_2 −→_β^∗ t_3” by (simp add: Beta star equals One star) qed This theorem completes the Church-Rosser proof.

Chapter 3

Nominal Reasoning Infrastructure

3.1 Preliminaries

Nominal Isabelle can be included in a theory using the usual Isabelle convention:

theory Foo imports Nominal begin . . . declarations, definitions and proofs . . . end The most important constants introduced in Nominal Isabelle are:

X-Symbol ASCII

permutation operation^• \< bullet > freshness ♯ \< sharp > support supp abstractions [ ].

Note that the symbol for freshness is different from Isabelle’s symbol # used for list-cons.

These constants are all polymorphic and therefore often need explicit typeannotations to be meaningful (see Appendix A.3). Lack of explicit typeannotations is usually what causes rules to not being applicable or to other non-intuitive behaviour. The meaning of these constants is described in more detail in the remaining sections of this chapter.

3.2 Atom Declarations

Before any nominal datatype can be defined, Nominal Isabelle needs to know which atom types will be used. Atom types stand for different kinds of (object) variables that might be bound in nominal datatypes.

Atom types can be declared using the command atom decl with the syntax

atom decl type_1. . . type_n

where type_1. . . type_n stand for some identifiers. A concrete example is

atom decl var tyvar

which might be used to formalise variables in lambda-terms and type-variables in simple types. After this declaration one can write

lemma foo: fixes x::”var” and T::”tyvar” . . .

to fix the types for the (Isabelle) variables x and T. Note that the atom declaration is a “global” declaration, in the sense that it can be issued only once. Therefore it needs to contain all atom types that are ever going to be used in a formalisation.

Each atom type contains countable infinitely many elements. They can be constructed by using the atom-type name as constructor and a natural number as argument. For example one can show:

lemma concrete atoms: shows ”var 0 ≠ var 1” by simp

However concrete atoms, such as var 0, are only rarely used in Nominal Isabelle. One example where they are used is to define the sets of “even” and “odd” atoms in the example file Support.thy.

When the atom declaration is issued, the Nominal Isabelle generates a number of lemmas that have been specialised according to the given atom types. One example is the lemma

lemma exists fresh^′:

assumes ”finite ((supp x)::var set)” shows ”∃ a::var. a ♯ x” . . .

which states that for every finitely supported x there exists an atom a of type var which is fresh for x (support and freshness will be explained in Section 3.4). If more than one atom type is declared, then the lemma exists fresh^′ will contain such a statement for every atom type. For example, in case we declare “atom decl var”, then exists fresh^′ will be the lemma

finite ((supp x)::var set) =⇒ ∃ a::var. a ♯ x

while in case of “atom decl var tyvar”, exists fresh^′ will be the collection of lemmas

finite ((supp x)::var set) =⇒ ∃ a::var. a ♯ x finite ((supp x)::tyvar set) =⇒ ∃ a::tyvar. a ♯ x

This kind of specialisation to concrete atom-types is provided automatically for lemmas that are frequently needed. For less frequently used lemmas only general versions are provided in the file Nominal.thy. Such general lemmas need to be specialised manually. For example the lemma

at exists fresh^′[OF at var inst]

is equivalent to

finite ((supp x)::var set) =⇒ ∃ a::var. a ♯ x

See also Section 3.6 for more details on the instantiation of generic lemmas.

3.3 Permutations

Permutations are bijective mappings from atoms to atoms. In Nominal Isabelle such permutations are represented as lists of atom-pairs. For this the type-abbreviation

‘x prm

standing for (‘x × ‘x) list has been introduced. The type variable ‘x can be instantiated with any declared atom type. For example one can write:

lemma another foo: fixes π_1::”var prm” and π_2::”tyvar prm” . . .

Since permutations are lists, they can be written using the standard notation for lists in Isabelle/HOL. Three examples permutations are

[(a, b)] [(a, a), (b, c)] []

where the first stands for the permutation that swaps the atoms a and b; the second swaps first b and c, and then a and a; the last stands for the identity permutation.

Two permutations can be composed by using list-append, written π_1 @ π_2. The inverse of a permutation is given by reversing the list. For this the standard reversal function, written (rev π), of Isabelle is used. Note however that (rev [(a,b)]) is not [(b,a)], but is equal to [(a,b)]:

lemma swap rev: fixes a b::”var” shows ”rev [(a,b)] = [(a,b)]” by simp

The infix operation^• that applies a permutation to an object has the polymorphic type

‘x prm ⇒ ‘a ⇒ ‘a.

One can write for example

lemma swap id: fixes a::”tyvar” shows ”[(a,a)]^• x = x” . . .

where the permutation [(a, a)] is applied to the object x, or

lemma perm bij: fixes π::”var prm” shows ”(rev π)^• (π^• x) = x” . . .

where π is applied to x, followed by an application of the inverse of π. For several standard types, Nominal Isabelle defines automatically permutation operations. Some examples are:

pairs: π^• (x, y) = (π^• x, π^• y) sets: π^• X = {π^• x |x. x ∈ X} lists: π^• [] = [] π^• (x # xs) = π^• x # π^• xs functions: π^• f = (λx. π^• f (rev π^• x)) natural numbers: π^• n = n booleans: π^• b = b

Permutation operations on atoms are defined such that

[]^• c = c

[(a,b)]#π^• c = b if π^• c = a a if π^• c = b π^• c otherwise

where the permutations has the same type as the atom c. In case the permutation is of different type than the atom they are acting on, then the permutation operation is defined as π^• a = a. For example one can show that

lemma perm ineq: fixes a::”var” shows ”∀ π::tyvar prm. π^• a = a” by (simp add: calc atm)

where the lemma calc atm contains all lemmas that are needed to analyse a permutation applied to an atom.

While the representation of permutations as lists is very convenient for composing two permutations and for building the inverse of a permutation, it is not a unique representation. Nominal Isabelle therefore defines the relation ≜ that states when two permutations (necessarily having the same type) are equal, namely:

constdefs prm eq :: ”‘x prm ⇒ ‘x prm ⇒ bool” (” ≜ ” [80,80] 80) ”π_1 ≜ π_2 ≡ ∀ a::‘x. π_1^•a = π_2^•a”

The tactic perm simp can be used to analyse many instances^1 of permutation operations applied to an arbitrary object. Its syntax is similar to the standard simp-tactic. For example one can add lemmas to, or delete from, the simplifier by issuing

(perm simp add: . . . del: . . . )

Three examples of equations which perm simp solves are:

lemma bar: fixes a b::”var” and l::”var list” and π::”var prm” shows ”[(a,b)]^•a = b” and ”[(a,b)]^•[(a,b)]^•l = l” and ”π^•(rev π)^•l = l” by perm simp+

1Unfortunately it is not known whether a suitable equational theory involving permutations applied to arbitrary objects is decidable.

3.4 Support

The most interesting feature of the nominal work is the notion of support (see Pitts [2003]). This notion corresponds roughly to the usual notion of what the set of free variables of an object is, however it is more general and this generality is crucial in a number of places in Nominal Isabelle. The definition of support is

constdefs supp :: ”‘a ⇒ (‘x set)” ”supp x ≡ {a . (infinite {b . [(a,b)]^•x ≠ x})}”

where the right-hand side stands for a set of atoms (in this case of type ‘x). Note that this definition is different from some other definitions one can find in the nominal literature.

For finitary structures, such as pairs, lists, lambda-terms, etc, the notion of support coincides with the usual notion of free variables. For example, assuming that the atom types var and tyvar are declared, then the support of a var-atom is

lemma supp atm: fixes a::”var” shows ”(supp a) = ({a}::var set)” and ”(supp a) = ({}::tyvar set)”

. . .

Note that the in the first statement the support of “a” is calculated with respect to the atom type var, while in the second with respect to tyvar. Because of the typing constraints for the constant supp, it is impossible in Nominal Isabelle to express in a single set what the support of an object is with respect to all atom types (the reason is that the set ({a}::var set) ∪ ({}::tyvar set) is ill-typed).

For pairs one can easily show that their support is equal to the support of their components (see Figure 3.1 for the calculation in Isar). The support of an empty list and list-cons is

supp [] = {} supp (x # xs) = supp x ∪ supp xs

Natural numbers and booleans have empty support:

supp n = {} supp b = {} lemma supp prod: 11 shows ”supp (x_1,x_2) = (supp x_1) ∪ (supp x_2)” 12 proof - 13 have ”supp (x_1,x_2) = {a. infinite {b. [(a,b)]^•(x_1,x_2) ≠ (x_1,x_2)}}” by (simp only: supp def) 14 also have ”. . . = {a. infinite {b. ([(a,b)]^•x_1,[(a,b)]^•x_2) ≠ (x_1,x_2)}}” by simp 15 also have ”. . . = {a. infinite {b. [(a,b)]^•x_1 ≠ x_1 ∨ [(a,b)]^•x_2 ≠ x_2}}” by simp 16 also have ”. . . = {a. infinite ({b. [(a,b)]^•x_1 ≠ x_1} ∪ {b. [(a,b)]^•x_2 ≠ x_2})}” 17 by (simp only: Collect disj eq) 18 also have ”. . . = {a. infinite {b. [(a,b)]^•x_1 ≠ x_1} ∨ infinite {b. [(a,b)]^•x_2 ≠ x_2}}” by simp 19 also have ”. . . = {a. infinite {b. [(a,b)]^•x_1 ≠ x_1}} ∪ {a. infinite {b. [(a,b)]^•x_2 ≠ x_2}}” 20 by (simp only: Collect disj eq) 21 also have ”. . . = (supp x_1) ∪ (supp x_2)” by (simp only: supp def) 22 finally show ”supp (x_1,x_2) = (supp x_1) ∪ (supp x_2)” . 23 qed 24 Figure 3.1: A proof for calculating the support of a pair: Line 6 uses the property when a pair is not equal, Line 9 uses the property when a union of two sets is infinite; Lines 7–8 and 10–11 transform a disjunction inside a collection into a union of two sets.

The support of an abstraction, a construct that will be described in more detail in Sec. 3.7, is

supp ([a].x) = supp x - supp a

However this equation has the proviso that x must have finite support. The support of a finite set, say X, is the union of the support of every element in X, that is

lemma supp fin set: assumes ”finite X” shows ”(supp X) = ( ⋃ x∈X. (supp x))” . . .

More complicated is the situation for infinitary structures, such as infinite sets and functions. For example it can be easily shown that the set of all atoms of type var has empty support (the set of all atoms of type var is in Isabelle represented by UNIV::var set). We first show that swapping two atoms in this set leaves the set unchanged:

lemma swap UNIV var: fixes a b::”var” shows ”[(a,b)]^•(UNIV::var set) = UNIV” by (auto simp add: perm set eq calc atm)

Now it is straightforward to conclude that UNIV::var set has empty support.

lemma supp UNIV var:

shows ”supp (UNIV::var set) = ({}::var set)” by (simp add: supp def swap UNIV var)

However, there are also objects in the HOL-logic that have infinite support. In the example file Support.thy it is shown that the set of “even”, respectively “odd”, atoms have infinite support. Pitts [2006] gives an example for a function that has infinite support (see Example 3.4 in Pitts [2006]).

3.5 Freshness

Derived from the concept of support is the notion of freshness (again for more background information see Pitts [2003]). Freshness is defined as

constdefs

fresh :: ”‘x ⇒ ‘a ⇒ bool” (” ♯ ” [80,80] 80) ”a ♯ x ≡ a /∈ supp x”

Recall that there are countable infinitely many elements in an atom type. Consequently if a given object x has finite support, then there must exist an atom outside the support of x; that is there must be an atom that is fresh for x. This fact is proved in the lemma exists fresh^′

lemma exists fresh^′:

assumes ”finite ((supp x)::var set)” shows ”∃ c::var. c ♯ x” . . .

Note that this lemma contains such facts for all declared atom types.

For two atoms of the same type the notion of freshness coincides with inequality:

lemma fresh atm: fixes a b::”var” shows ”(a ♯ b) = (a ≠ b)” . . .

Unwinding the definition of support for other types gives the following facts:

pairs: a ♯ (x, y) = (a ♯ x ∧ a ♯ y) sets: a ♯ {} [[pt TYPE(‘a) TYPE(‘b); at TYPE(‘b)]] =⇒ a ♯ {x} = a ♯ x lists: a ♯ [] a ♯ (x # xs) = (a ♯ x ∧ a ♯ xs) natural numbers: a ♯ n booleans: a ♯ b

The crucial property of freshness is that if two atoms are fresh for an object, then swapping those atoms leaves x unchanged, that is

lemma perm fresh fresh: fixes a b::”var” assumes ”a ♯ x” and ”b ♯ x” shows ”[(a,b)]^•x = x” . . .

In order to keep formulae in goals manageable, one often wants to switch between a tupled version of freshness and an equivalent un-tupled series of freshness-conjunctions. For this the built-in simplifier has been set up so that it can solve automatically goals such as:

lemma fresh tuples: fixes a::”var” shows ”a ♯ (t_1,t_2) =⇒ a ♯ t_1 ∧ a ♯ t_2” and ”[[a ♯ t_1; a ♯ t_2]] =⇒ a ♯ (t_1,t_2)” and ”a ♯ (t_1,t_2) =⇒ a ♯ t_2” and ”a ♯ (t_1,t_2) =⇒ a ♯ (t_2,t_1,t_1,t_2)” by simp all

Although the reasoning infrastructure in Nominal Isabelle is designed to minimise the need of generating fresh atoms, sometimes there is no way around generating one. To facilitate this, the tactic generate fresh has been introduced. For example a fresh atom with type var is generated by applying the tactic

apply(generate fresh ”var”)

This tactic will scan for all variables in the current goal and then introduce a new assumption with an atom being fresh for all the variables for which the tactic can determine that they have finite support.

3.6 Permutation Types and Finitely Supported Types

The file Nominal.thy is a library containing general facts about support and freshness. These facts depend on the permutation operation to behave properly on every type it acts on. In order to make precise what behaving properly means, the following predicate is defined in Nominal.thy:

pt TYPE(‘a) TYPE(‘x) ≡ ∀ x. []^• x = x ∧ ∀ π_1 π_2 x. (π_1 @ π_2)^• x = π_1^• π_2^• x ∧ ∀ π_1 π_2 x. π_1 ≜ π_2 −→ π_1^• x = π_2^• x

(3.1)

This predicate takes two types as arguments (indicated by the keyword TYPE^2): the first type is an arbitrary type over which the permutaion acts and the second is an atom-type over which permutations are built. These arguments impose the following type-constraints on the right-hand side: in the first clause x has type ‘a and the empty list is of type ‘x prm; in the second and third clause x has type ‘a, and π_1 and π_2 have type ‘x prm. The type argement of pt are necessary because HOL does not allow any free type-variables on the right-hand sides of definitions.

The idea of the predicate pt is that a type ‘a is a permutation type with respect to the atom-type ‘x provided the identity permutation (empty list) leaves all elements of ‘x unchanged, that a composition of two permutations can be “un-composed” and that the application of two equal permutations has to produce the same result (see Page 23 for the definition of permutation equality). These three properties are sufficient to infer most facts about support and freshness.

The predicate pt is used, for example, in the proof of the fact about all swappings leaving an object unchanged implies that also all permutations leave this object unchanged.

lemma pt swap eq aux: fixes y :: ”‘a” assumes ”pt TYPE(‘a) TYPE(‘x)” and ”∀ (a::‘x) (b::‘x). [(a,b)]^•y = y” shows ”∀ π::‘x prm. π^•y = y” . . .

Nominal.thy proves a number of lemmas which show that pt holds “hereditarily” through the type-structure. For example

lemma shows pt unit inst: ”pt TYPE(unit) TYPE(‘x)” and pt bool inst: ”pt TYPE(bool) TYPE(‘x)” and pt list inst: ”pt TYPE(‘a) TYPE(‘x) =⇒ pt TYPE(‘a list) TYPE(‘x)” and pt prod inst: ”[[pt TYPE(‘a) TYPE(‘x); pt TYPE(‘b) TYPE(‘x)]] =⇒ pt TYPE(‘a × ‘b) TYPE(‘x)” . . .

Such lemmas allow one to lift the above lemma pt swap eq aux to apply to ‘a lists, for example. Resolving pt swap eq aux with pt list inst by writing

pt swap eq aux[OF pt list inst]

gives the lemma

2See Section 2.3.2 in Wenzel [2007] for more deatils about this construction.

lemma pt swap eq aux for lists: fixes y :: ”‘a list” assumes ”pt TYPE(‘a) TYPE(‘x)” and ”∀ (a::‘x) (b::‘x). [(a,b)]^•y = y” shows ”∀ π::‘x prm. π^•y = y” . . .

where the difference is that y has type ‘a list. Such lifting of lemmas is automated in Nominal Isabelle employing axiomatic type classes (which are described in Wenzel [2004]). This mechanism is explained next.

The atom declaration generates for each atom type two axiomatic typeclasses, which state the conditions for permutation types and finitely supported types. These type-classes are named pt type_i and fs type_i where type_i ranges over the atom types introduced by atom decl. Nominal Isabelle derives a number of instance proofs so that Isabelle’s type-system can in most circumstances automatically infer when a type is a permutation type and/or a finitely supported type. Nominal datatypes are always permutation types and their elements are always finitely supported, see Chapter 4. Sets and functions are permutation types, but not finitely supported types (there are some infinite sets and some functions that do not have finite support).

3.7 Abstractions and Alpha-Equivalence

As we shall see in Chapter 4 nominal datatypes represent alpha-equivalence classes.

Chapter 4

Nominal Datatypes

4.1 Declaration

In contrast to standard datatypes of Isabelle/HOL, nominal datatypes represent alpha-equivalence classes. They can be declared using the nominal datatype keyword. Its syntax is similar to the standard datatype declaration, except it allows one to specify where binders occur using ≪ ≫ . For example alpha-equated lambda-terms can be specified as:

nominal datatype lam = Var ”var” | App ”lam” ”lam” | Lam ”≪var≫lam”

assuming that var has been declared as an atom type. The constructor Lam is then treated as a constructor with two arguments: a var and a lam. This means one can write Lam a t. However, it is often more convenient to introduce special syntax annotations for constructors with binders. One example of such an annotation is

nominal datatype lam = Var ”var” | App ”lam” ”lam” | Lam ”≪var≫lam” (”Lam [ ]. ” [100,100] 100)

which allows one to write, instead of Lam a t, the more memorable Lam [a].t for lambda-abstractions. (Note that λ is already in use by Isabelle and cannot be used in special syntax.) However this kind of direct syntax annotation does not work in situations when one wants to change the order of arguments. For example the lambda-calculus with let can be specified as nominal datatype lam = Var ”var” | App ”lam” ”lam” | Lam ”≪var≫lam” (”Lam [ ]. ” [100,100] 100) | Let ”≪var≫lam” ”lam”

To obtain the usual notation for Lets, one can use the following abbreviation:

abbreviation LetBe :: ”var⇒lam⇒lam⇒lam” (”Let be in ” [100,100,100] 100) where ”Let x be t_1 in t_2 ≡ lam.Let x t_2 t_1”

Nested binders can be specified as shown in the following nominal datatype where the first argument in ImpR abstracts over a name and a coname.

nominal datatype trm = Ax ”name” ”coname” | Cut ”≪coname≫trm” ”≪name≫trm” | ImpR ”≪name≫≪coname≫trm” ”coname” | ImpL ”≪coname≫trm” ”≪name≫trm” ”name”

Currently the following restrictions are in place for nominal datatypes:

• inside ≪ ≫ only one atom type is allowed

• no type variables (that is ‘a and so on) are allowed in nominal datatype definitions

• no nested types are allowed; they need to be explicitly defined as mutual recursive datatype, and

• no function types can be used (the reason is that it is generally too difficult to ensure automatically that functions have finite support).

The first and third restriction seem in practice the most inconvenient ones and the hope is that they can be lifted in future versions of Nominal Isabelle. This is desirable, because at the moment type-schemes from Hindley-Milner typing algorithm, for example, need to be specified as

nominal datatype ty = TVar ”var” | Fun ”ty” ”ty” (” → ” [100,100] 100) nominal datatype tyS = Ty ”ty” | All ”≪var≫tyS” (”∀ [ ]. ” [100,100] 100) instead of abstracting a list or set of names. Also the third restriction is annoying: while with standard datatypes one can declare

datatype trm = Var ”var” | Fun ”trm list”

this is not yet possible with nominal datatypes. Instead, the trm-lists need to be explicitly unfolded like

nominal datatype trm = Var ”var” | Fun ”trm list” and trm list = MyNil | MyCons ”trm” ”trm list”

This is quite awkward and can at the moment only be circumvented in cases the nominal datatype has no binders. In such cases one can declare the datatype using the normal datatype-mechanism and manually provide a permutation operation for this datatype (see Section ?? for further details).

For nominal datatypes a number of lemmas are automatically generated. The most important are named

nominal-datatype-name.perm

nominal-datatype-name.supp

nominal-datatype-name.fresh

nominal-datatype-name.inject

In case of the lambda-calculus they are defined as follows:

lemma lam.perm: fixes π::”var prm” shows ”π^• Var x = Var (π^• x)” and ”π^• App t_1 t_2 = App (π^• t_1) (π^• t_2)” and ”π^• Lam [x].t = Lam [(π^• x)].(π^• t)”

lemma lam.supp:

shows ”supp (Var x) = (supp x)” and ”supp (App t_1 t_2) = (supp t_1) ∪ (supp t_2)” and ”supp (Lam [x].t) = (supp ([x].t))” lemma lam.fresh:

shows ”a ♯ (Var x) = a ♯ x” and ”a ♯ (App t_1 t_2) = (a ♯ t_1 ∧ a ♯ t_2)” and ”a ♯ (Lam [x].t) = (a ♯ [x].t)”

lemma lam.inject:

shows ”(Var x = Var y) = (x = y)” and ”(App t_1 t_2 = App s_1 s_2) = (t_1 = s_1 ∧ t_2 = s_2)” and ”(Lam [x].t_1 = Lam [y].t_2) = ([x].t_1 = [y].t_2)”

While the first three lemmas are added to the simplifier, please note that for nominal datatypes the lemma nominal-datatype-name.inject is not automatically added to the simplifier (unlike for standard datatypes).

4.2 Strong Structural Inductions

Nominal Isabelle provides for every nominal datatype a “weak” induction principle, whose name is nominal-datatype-name.induct. In case of the lambdacalculus it is as follows: ⋀ x. P (Var x) ⋀ t_1 t_2. [[P t_1; P t_2]] =⇒ P (App t_1 t_2) ⋀ x t. P t =⇒ P (Lam [x].t) P t (4.1)

This induction principle can be used to show, for example, that swapping the atom a with itself behaves like an identity on lambda terms, see Figure 4.1 for a detailed Isar-proof of this fact.

This induction principle is “weak”, because one has to prove the lambdacase for all binders x (note the meta-quantification ⋀ x in (4.1)). While this is no problem in the proof shown in Figure 4.1, in more complicated circumstances it means that one has to find a suitable renaming to get the Lam-case through. In order to reduce the amount of effort in such circumstances, strong induction principles are automatically generated for nominal datatypes. In case of the lambda-calculus the strong induction principle lemma swap id: 1 fixes a::”var” and t::”lam” 2 shows ”[(a,a)]^• t = t” 3 proof (induct t rule: lam.induct) 4 case (Var x) 5 show ”[(a,a)]^• Var x = Var x” by (simp add: calc atm) 6 next 7 case (App t_1 t_2) 8 have ih_1: ”[(a,a)]^• t_1 = t_1” and ih_2: ”[(a,a)]^• t_2 = t_2” by fact+ 9 then show ”[(a,a)]^• App t_1 t_2 = App t_1 t_2” by simp 10 next 11 case (Lam x t) 12 have ih: ”[(a,a)]^• t = t” by fact 13 then show ”[(a,a)]^• Lam [x].t = Lam [x].t” by (simp add: calc atm) 14 qed 15 Figure 4.1: A simple proof illustrating the use of the “weak” induction principle lam.induct: the proof shows that swapping an atom a with itself leaves lambda-terms unchanged.

(called lam.induct) is as follows: ⋀ x c. P c (Var x) ⋀ t_1 t_2 c. [[ ⋀ c‘. P c‘ t_1; ⋀ c‘. P c‘ t_2]] =⇒ P c (App t_1 t_2) ⋀ x t c. [[x ♯ c; ⋀ c‘. P c‘ t]] =⇒ P c (Lam [x].t)

P c t

(4.2)

where c stands for the freshness context of the induction. The precondition x ♯ c in the third case implies that one has to prove the lambda-case for only those binders x that are fresh with respect to this context. This usually makes the binder cases as simple as the informal reasoning where the variable convention is employed.

Before we give an example, we shall describe how strong structural inductions can be used (the usual method induct does not work with them). To apply a strong structural induction principle one has to use the method nominal induct, which applies a strong induction rule and helps with instantiating the freshness context. Its syntax is

(nominal induct x_1. . .x_n arbitrary: y_1. . .y_m avoiding: z_1. . .z_k

rule: name-of-the-strong-induction-rule) where arbitrary and avoiding are optional. The meaning of the variables are:

• the x_i stand for the variables over which the induction is done (in structural inductions this is only a single variable),

• the y_j stand for the variables that are generalised in the induction, and

• the z_l stand for the variables in the freshness context of the nominal induction—this context describes what is usually meant be the variable convention.

For example when proving the usual substitution lemma^1 in the lambdacalculus

lemma substitution lemma: assumes a: ”x ≠ y” and b: ”x ♯ L” shows ”M[x::=N][y::=L] = M[y::=L][x::=N[y::=L]]”

the induction proceeds over the structure of M (no generalisation is needed for the induction to go through). If one follows the proof of this lemma given in Barendregt [1981], then in the lambda-case it is assumed that the binder, say z, has to satisfy the following constraints: z ≠ x, z ≠ y, z ♯ N and z ♯ L. These assumptions are usually justified by the variable convention. To formalise Barendregt’s proof, one can set up the induction using:

(nominal induct M avoiding: x y N L rule: lam.induct)

In the lambda-case one can then assume the constraints from the variable convention. This means we can formalise the proof of the substitution lemma roughly as follows:

lemma substitution lemma: 1 assumes a: ”x ≠ y” and b: ”x ♯ L” 2 shows ”M[x::=N][y::=L] = M[y::=L][x::=N[y::=L]]” 3 using a b 4 proof (nominal induct M avoiding: x y N L rule: lam.strong induct) 5 case (Var z) . . . 6 next 7 case (App M_1 M_2) . . . 8 next 9 case (Lam z M_1) 10 have vc: ”z ♯ x” ”z ♯ y” ”z ♯ N” ”z ♯ L” by fact 11 . . . 12 qed 13 1The substitution function for the nominal datatype lam will be defined in Section 4.4.

where Line 11 contains the freshness assumptions about the binder z. With those assumptions the proof of the substitution lemma is quite simple and does not need any renaming of binders—just like the proof of Barendregt.

4.3 Equivariance Lemmas

An important concept in the nominal logic work is equivariance [Pitts, 2003]. In Nominal Isabelle, definition of functions and inductive definition (described in the next sections) rely on this concept. Therefore a special attribute, called [eqvt], has been introduced for indicating when a lemma establishes the equivariance of a function or a relation. A function f is equivariant provided

π^•(f x_1. . .x_n) = f (π^•x_1). . .(π^•x_n)

holds where the x_i stand for all the arguments of f. A relation R is equivariant provided

if R x_1. . .x_n then also R (π^•x_1). . .(π^•x_n)

holds. Again the x_i stand for all the arguments of R. In order to make such lemmas known to the system, they can be tagged with the attribute [eqvt]. Two examples are

lemma substitution is equivariant[eqvt]: fixes π::”var prm” shows ”π^•(t_1[x::=t_2]) = (π^•t_1)[(π^•x)::=(π^•t_2)]” . . .

lemma equality over lam is equivariant[eqvt]: fixes π::”var prm” and s t::”lam” assumes ”s = t” shows ”(π^•s) = (π^•t)” . . .

When the [eqvt]-attribute is attached to a lemma, the system checks that certain conditions are satisfied. In case of functions, where the equivariance lemma must be of the form lhs = rhs, the conditions state that the lefthand side must be of the form π^• expr and the right-hand side must be equal to expr except that every free variable, say x, in it is replaced by π^• x (see lemma substitution is equivariant above). In case of relations the conclusion must be equal to the premise, except again that the permutation π acts on every free variable in the conclusion. When the conditions are not met then the following error message is displayed at the end of the proof:

*** Error in attribute "Nominal.eqvt":
*** "offending lemma" does not comply with the
    form of an equivariance lemma ...

A number of lemmas that are generated by Nominal Isabelle get automatically tagged with the [eqvt]-attribute. Some of them are listed in Appendix B.1. Also all constructors of a nominal datatype have equivariance lemmas. For example for lambda-terms we have:

lemma lam.perm[eqvt]: fixes π::”var prm” shows ”π^• Var x = Var (π^• x)” and ”π^• App t_1 t_2 = App (π^• t_1) (π^• t_2)” and ”π^• Lam [x].t = Lam [(π^• x)].(π^• t)”

The equivariance lemmas are collected under the theorem called eqvts. Therefore one can use them in proofs using the standard proving tools. For example

• (simp add: eqvts)

• (rule eqvts)

The crucial point of equivariance is that an equivariant function (similar relations) have empty support. This fact will play a crucial role in the next section.

4.4 Function Definitions

To be done.

nominal primrec subst :: ”lam⇒var⇒lam⇒lam” (” [ ::= ]” [100,100,100] 100) where ”(Var x)[y::=t‘] = (if x=y then t‘ else (Var x))” | ”(App t_1 t_2)[y::=t‘] = App (t_1[y::=t‘]) (t_2[y::=t‘])” | ”x ♯ (y,t‘) =⇒ (Lam [x].t)[y::=t‘] = Lam [x].(t[y::=t‘])” apply(finite guess)+ apply(rule TrueI)+ apply(simp add: abs fresh) apply(fresh guess)+ done

4.5 Inductive Definitions and Strong Rule Induction Principles

Inductive definition can be declared in Isabelle using the inductive mechanism. If such a definition contains nominal datatypes, then one often needs a stronger induction principle that has the variable convention built in. The strengthening can be achieved using the command nominal inductive. While the strengthening is completely automatic in case of structural inductions (see Sec. 4.2), there is in some work needed from the user in case of strong inductions for inductive definitions. This is because the strengthening relies on three facts

• the inductive definition has to be equivariant and

• every rule of an inductive definition has to be formulated so that no variable standing for a binder occurs in the support of this rule, and

• if there are more than one binder in a rule, then these binders also need to be distinct.

While the first property can often be automatically derived using the command equivariance, the latter two require some proof obligations to be discharged.

As an example consider the inductive definition for the typing relation in the simply-typed lambda-calculus. The nominal datatypes for terms and types are defined as:

nominal datatype lam = Var ”var” | App ”lam” ”lam” | Lam ”≪var≫lam” (”Lam [ ]. ” [100,100] 100)

nominal datatype ty = TVar ”string” | TArr ”ty” ”ty” (” → ” [100,100] 100)

Typing contexts will be represented as lists of (var,ty)-pairs. We need to ensure that a var in a context has only one associated type. This can be achieved with a validity predicate defined inductively as:

inductive valid :: ”(var×ty) list ⇒ bool” where [intro]: ”valid []” | [intro]: ”[[valid Γ; x ♯ Γ]]=⇒ valid ((x,T)#Γ)”

The typing judgement can then be defined by the rules:

inductive typing :: ”(var×ty) list⇒lam⇒ty⇒bool” (” ⊢ : ” [60,60,60] 60) where ty var[intro]: ”[[valid Γ; (x,T) ∈ set Γ]] =⇒ Γ ⊢ (Var x) : T” | ty app[intro]: ”[[Γ ⊢ t_1 : T_1→T_2; Γ ⊢ t_2 : T_1]] =⇒ Γ ⊢ App t_1 t_2 : T_2” | ty lam[intro]: ”[[x ♯ Γ; (x,T_1)#Γ ⊢ t : T_2]] =⇒ Γ ⊢ Lam [x].t : T_1→T_2”

With this inductive definition comes automatically an induction principle, called typing.induct. However, as said before, this induction principle is too weak for establishing any non-trivial fact about typing. In order to strengthen it, we need to make sure that the typing relation is equivariant, that means the property

Γ ⊢ t : T =⇒ π^• Γ ⊢ π^• t : π^• T

holds. For this, we have to make sure that every side-condition in the definition of typing is equivariant. That means, for example, for the rule ty var that valid, pairing, ∈, set and Var must be equivariant. Nominal Isabelle knows that the latter four are equivariant, but not that the user-defined predicate valid is. Therefore attempting to prove equivariance of typing using the command equivariance will fail: the statement

equivariance typing

results in the error message

*** Could not prove equivariance for introduction rule
*** [[valid Γ; (x, T) ∈ set Γ]] =⇒ Γ ⊢ Var x : T
     ...

where the second line indicates the rule, which causes equivariance to fail (in this case ty var). To solve the problem equivariance of valid needs to be proved first. This can done by stating

equivariance valid

This succeeds because [], ♯, # and pairing, which are used in the definition of valid, are all equivariant. Having equivariance of valid at the disposal, the statement

equivariance typing succeeds.

Now we can strengthen the induction principle for the typing relation using nominal inductive. Issuing

nominal inductive typing

will open a proof requiring to show the following three goals:

[[x ♯ Γ; (x, T_1) # Γ ⊢ t : T_2 ]] =⇒ x ♯ Γ [[x ♯ Γ; (x, T_1) # Γ ⊢ t : T_2 ]] =⇒ x ♯ Lam [x].t [[x ♯ Γ; (x, T_1) # Γ ⊢ t : T_2 ]] =⇒ x ♯ T_1→T_2

These goals correspond to the rule ty lam of the typing-relation (the only rule where a binder occurs). They ensure that the binder x does not appear freely in the conclusion of ty lam, i.e. Γ ⊢ Lam [x].t : T_1→T_2 . The first one is implied by the side-condition x ♯ Γ in ty lam; the second holds because x is bound in Lam [x].t; the last one holds since vars cannot appear in types. However we need a little side-lemma which establishes this last fact:

lemma fresh ty: fixes x::”var” and T::”ty” shows ”x ♯ T” by (nominal induct T rule: ty.strong induct) (simp all add: fresh string)

Armed with this side-lemma, we can complete the proof opened by nominal inductive as follows:

nominal inductive typing by (simp all add: fresh ty abs fresh)

This results in the following strong induction principle, called typing.strong induct, being proved:

Γ ⊢ t : T [[valid Γ; (x, T) ∈ set Γ]] =⇒ P c Γ (Var x) T [[Γ ⊢ t_1 : T_1→T_2; ⋀ c‘. P c‘ Γ t_1 (T_1→T_2); Γ ⊢ t_2 : T_1; ⋀ c‘. P c‘ Γ t_2 T_1]] =⇒ P c Γ (App t_1 t_2) T_2 [[x ♯ c; x ♯ Γ; (x,T_1)#Γ ⊢ t : T_2; ⋀ c‘. P c‘ ((x,T_1)#Γ) t T_2]] =⇒ P c Γ (Lam [x].t) (T_1→T_2) P c Γ t T

We will show next that this stronger version of the induction principle is quite useful for formalising proofs that use the variable convention. Like in the case of the strong structural induction principles for nominal datatypes, we can instantiate the freshness context c of the induction so that the binder x in the lambda-case is fresh with respect to this context. Consider Figure 4.2 where the weakening lemma is proved. In the lambda-case of this proof the fact that the binder x is fresh for the typing-context Γ_2 is needed—otherwise we would have to do explicit renamings as the rule ty lam requires that x ♯ Γ_2 and we also cannot establish that valid ((x, T_1) # Γ_2) holds. However we can establish these facts by stating in nominal induct that Γ_2 should be avoided.

A disadvantage of nominal inductive is that the automatic proof often requires that one states the rules for the inductively defined predicate slightly different than one is used to from “pencil-and-paper” reasoning. Recall that one requirement for the automatic strengthening is that no binder should not be in the upport of the conclusion of a rule. Now consider the following definition of beta-reduction:

inductive beta :: ”lam⇒lam⇒bool” (” −→_β ” [80,80] 80) where AppL: ”s_1 −→_β s_2 =⇒ App s_1 t −→_β App s_2 t” | AppR: ”s_1 −→_β s_2 =⇒ App t s_1 −→_β App t s_2” | Lam: ”s_1 −→_β s_2 =⇒ Lam [x].s_1 −→_β Lam [x].s_2” | Beta: ”App (Lam [x].s_1) s_2 −→_β s_1[x::=s_2]”

The rules lam and beta contain the binder x and in order to strengthen the induction principle in both cases the x needs to be fresh for the rules’ conclusion. When stating

nominal inductive beta

we are required to discharge the following four proof-obligations

s_1 −→_β s_2 =⇒ x ♯ Lam [x].s_2 s_1 −→_β s_2 =⇒ x ♯ Lam [x].s_2 x ♯ App (Lam [x].s_1) s_2 x ♯ s_1[x::=s_2]

where we can easily discharge the first two (so rule lam is ok), but the last two are unprovable: x is fresh for Lam [x].s_1, but it may occur in s_2. As for the term s_1[x::=s_2], we know that any occurrences of x in s_1 will be masked by the substitution, but x may again be free in s_2. To obtain automatically a strong induction principle for beta-reduction, it must be defined as:

inductive beta :: ”lam⇒lam⇒bool” (” −→_β ” [80,80] 80) where abbreviation 1 sub ctxt :: ”(var×ty) list⇒(var×ty) list⇒bool” (” ⊆ ” [60,60] 60) 2 where 3 ”Γ_1 ⊆ Γ_2 ≡ ∀ x T. (x,T) ∈ set Γ_1 −→ (x,T) ∈ set Γ_2” 4 5 lemma weakening: 6 fixes Γ_1::”(var×ty) list” 7 and t ::”lam” 8 and T ::”ty” 9 assumes asms: ”Γ_1 ⊢ t : T” ”valid Γ_2” ”Γ_1 ⊆ Γ_2” 10 shows ”Γ_2 ⊢ t : T” 11 using asms 12 proof (nominal induct avoiding: Γ_2 rule: typing.strong induct) 13 case (ty var Γ_1 x T) (variable case) 14 have ”Γ_1 ⊆ Γ_2” by fact 15 moreover 16 have ”valid Γ_2” by fact 17 moreover 18 have ”(x,T) ∈ set Γ_1” by fact 19 ultimately show ”Γ_2 ⊢ Var x : T” by auto 20 next 21 case (ty lam x Γ_1 T_1 t T_2) (lambda case) 22 have vc: ”x ♯ Γ_2” by fact 23 have ih: ” ⋀ Γ_3. [[valid Γ_3; (x,T_1)#Γ_1 ⊆ Γ_3]] =⇒ Γ_3 ⊢ t : T_2” by fact 24 have ”Γ_1 ⊆ Γ_2” by fact 25 then have ”(x,T_1)#Γ_1 ⊆ (x,T_1)#Γ_2” by simp 26 moreover 27 have ”valid Γ_2” by fact 28 then have ”valid ((x,T_1)#Γ_2)” using vc by auto 29 ultimately have ”(x,T_1)#Γ_2 ⊢ t : T_2” using ih by simp 30 with vc show ”Γ_2 ⊢ Lam [x].t : T_1→T_2” by auto 31 qed (auto) (application case is automatic) 32 Figure 4.2: The Isar-proof for the weakening lemma. In Lines 1-4 the notion of a sub-context is defined for lists. The interesting case in the weakening lemma (Lines 22-31) crucially depends on the assumption that x ♯ Γ_2 holds (Line 23). This fact is used in Lines 29 and 31. It is obtained by using the strong induction principle for the typing relation and setting up the induction so that it avoids Γ_2 (Line 13).

AppL: ”s_1 −→_β s_2 =⇒ App s_1 t −→_β App s_2 t” | AppR: ”s_1 −→_β s_2 =⇒ App t s_1 −→_β App t s_2” | Lam: ”s_1 −→_β s_2 =⇒ Lam [x].s_1 −→_β Lam [x].s_2” | Beta: ”x ♯ s_2 =⇒ App (Lam [x].s_1) s_2 −→_β s_1[x::=s_2]”

where in the last clause the freshness constraint x ♯ s_2 is added. With this additional constraint, we are able to use nominal inductive to generate the strong induction principle beta.strong induct.

nominal inductive beta by (simp all add: abs fresh fresh subst)

Adding the constraint, however, is annoying because both versions of beta can be shown to define the same reduction relation. While it is prossible to use the simpler version of beta and then derive manually a strong induction principle, the possibility of having the strong induction principle automatically derived by nominal inductive often outweighs the benefits of being word-for-word faithful to the “pencil-and-paper” definitions. If one really prefers the simpler version for the beta-rule, then one can prove the following lemma which alpha-converts the terms before applying the more restricted version of the beta-rule:

lemma better beta intro:

shows ”App (Lam [x].s_1) s_2 −→_β s_1[x::=s_2]” proof - obtain x‘::”var” where fs: ”x‘ ♯ (x,s_1,s_2)” by (rule exists fresh, rule fin supp, blast) have ”App (Lam [x].s_1) s_2 = App (Lam [x‘].([(x‘,x)]^•s_1)) s_2” using fs by (rule tac sym, perm simp add: lam.inject alpha fresh atm fresh prod) also have ”. . . −→_β ([(x‘,x)]^•s_1)[x‘::=s_2]” using fs by (simp add: beta.Beta) also have ”. . . = s_1[x::=s_2]” using fs by (simp add: subst rename) finally show ”App (Lam [x].s_1) s_2 −→_β s_1[x::=s_2]” by simp qed where we need the side-lemma

lemma subst rename: assumes a: ”x‘ ♯ s_1” shows ”([(x‘,x)]^•s_1)[x‘::=s_2] = s_1[x::=s_2]” using a by (nominal induct s_1 avoiding: x x‘ s_2 rule: lam.strong induct) (auto simp add: calc atm fresh atm abs fresh)

Nominal inductive has one advanced feature that has not yet shown up in the given examples: in some cases one wants to strengthen over a variable which is not a binder. For example Crary [2005] defines the following types for lambda-terms:

nominal datatype ty =

TBase | TUnit | Arrow ”ty” ”ty” (” → ” [100,100] 100)

and then defines the following two mutually recursive relations

inductive alg equiv:: ”(var×ty) list⇒lam⇒lam⇒ty⇒bool” (” ⊢ ⇔ : ”) and alg path equiv:: ”(var×ty) list⇒lam⇒lam⇒ty⇒bool” (” ⊢ ↔ : ”) where QAT Base: ”[[s ⇓ p; t ⇓ q; Γ ⊢ p ↔ q : TBase ]] =⇒ Γ ⊢ s ⇔ t : TBase” | QAT Arrow: ”[[x ♯ (Γ,s,t); (x,T_1)#Γ ⊢ App s (Var x) ⇔ App t (Var x) : T_2]] =⇒ Γ ⊢ s ⇔ t : T_1 → T_2” | QAT One: ”valid Γ =⇒ Γ ⊢ s ⇔ t : TUnit” | QAP Var: ”[[valid Γ; (x,T) ∈ set Γ]] =⇒ Γ ⊢ Var x ↔ Var x : T” | QAP App: ”[[Γ ⊢ p ↔ q : T_1 → T_2; Γ ⊢ s ⇔ t : T_1]] =⇒ Γ ⊢ App p s ↔ App q t : T_2”

The interesting rule is QAT Arrow. Although no binder occurs in the definition at all, one wants in proofs by induction over this definition to be able to assume that the variable x is sufficiently fresh (not just fresh for Γ, s and t). Although not obvious, this is permitted because the x does not appear in the conclusion of that rule. The strengthening of the induction principle over this variable can be achieved by giving to nominal inductive the information that the strengthening should happen over x. For this one uses avoids. For example strengthening the induction principle for the relations above can be achieved by

equivariance alg equiv

nominal inductive alg equiv avoids QAT Arrow: x by (simp all add: fresh ty fresh prod)

Note that in mutual inductive definitions only one of the predicates needs to be mentioned in equivariance and nominal inductive. The strong induction principle called

alg equiv alg path equiv.strong inducts

(note the “s” at the end used in mutually inductive definitions) can now be used to do inductions over both relations.

Three technical points concerning nominal inductive should be noted: the strengthening only works with inductive definition that have been declared using inductive ( does not work); in case the strengthening does not require any subgoals to be discharged (this happens in the cases where the inductive definition does not contain any binders), then the opened proof needs to be closed by a dot, for example:

nominal inductive foo .

The general syntax for nominal inductive is

nominal inductive name-of-inductive-relation avoids rule-name_1: x_1 x_2 . . . and rule-name_2: y_1 y_2 . . . . . .

At the moment there is also one limitation placed upon nominal inductive: so far it cannot deal correctly with any logical connective in rules.

(Equivariance for different atom-types)

Chapter 5

Advanced Topics

5.1 Functions That Need to Generate Fresh Names

Sometimes functions explicitly depend on choosing fresh names. To abstract away from any particular choice of such names, Pitts introduced the function fresh fun (he called this function just fresh). This function takes as argument a function that is of type atom to ‘a. The main property of fresh fun is

if a ♯ (h, ha) then fresh fun h = h a

where the function h needs to be finitely supported. With this, one can effectively control when the choice of a fresh name is made in a proof. An example where fresh fun needs to be used is the CPS-translation of Plotkin:

nominal primrec CPS :: ”lam ⇒ lam” where ”CPS (Var x) = fresh fun (λk. Lam [k].(App (Var k) (Var x)))” | ”CPS (Lam [x].M) =

fresh fun (λk. Lam [k].(App (Var k) (Lam [x].(CPS M))))” | ”CPS (App M N) = fresh fun (λk. Lam [k].(App (CPS M) (fresh fun (λm. Lam [m].(App (CPS N) (fresh fun (λn. App (App (Var m) (Var n)) (Var k))))))))”

However, it is often a bit awkward to reason about fresh fun because one can only replace fresh fun h with a freshly chosen name, say a, if one can determine that a is fresh for the function h and the application h a (see above).

To simplify the reasoning involving fresh fun, the tactic fresh fun simp has been introduced. To use it, one has to first generate a fresh name using the generate fresh tactic and then can apply this tactic. It uses the generated fresh name to instantiate one instance of the fresh fun-term. In doing so, it attempts to solve all conditions required by the main property of fresh fun. If there are unsolved goals, then they are returned to the user. Using the option (no asm) one can direct the tactic to only search for instances of fresh fun in the conclusion of the current goal. Possible uses are:

• fresh fun simp

• fresh fun simp (no asm)

Chapter 6

Examples

Some examples are included in:

     [ISABELLE HOME]/src/HOL/Nominal/Examples:

Weakening.thy a proof for weakening in the simply-typed lambda-calculus
Fsub.thy System F with subtyping (Part 1A of the PoplMmark Chal-

lenge [Aydemir et al., 2005, PoplMark Challenge])

Lam Funs.thy contains the definitions of important functions in the

lambda-calculus

CR.thy Church-Rosser proof from Barendregt’s lambda-calculus

book [1981]

CR Takahashi.thy a much simpler proof for the Church-Rosser property taken

from the work of Takahashi [1995] and Pollack [1995]

CK Machine.thy formalisation of soundness and completeness for a CBV CK-

machine; also type-preservation and progress are shown

SN.thy strong normalisation from the Proofs and Types book by

Girard et al. [1989]

Height.thy a simple example suggested by D. Wang about the height of

lambda-terms

Lambda mu.thy datatype declaration for the lambda-mu calculus
Class.thy a verification of the main result from Urban [2000] about

cut-elimination in classical logic

SOS.thy some typical SOS-proofs about the simply-typed lambda-

calculus

Crary.thy a formalisation of the chapter on logical relations by Crary

[2005]

VC Compatible.thy gives two examples of faulty lemmas derived with the help

of the variable convention about binders

Support.thy includes some calculations of the support in non-trivial in-

stances

Contexts.thy shows the equivalence of beta-reduction defined in the

Plotkin-style and Felleisen-Hieb-style

Type Preservation.thy a type preservation proof for beta-reduction in the simply-

typed lambda-calculus

Standardization.thy a standardization proof by Matthes formalised by Berghofer

Appendix A

Frequently Asked Questions

A.1 The atom decl command does not work.

If you get the error message:

*** Outer syntax error: end of input expected,
*** but identifier "atom_decl" was found

while stepping through a theory-file, then the most likely problem is that the key-word file for Nominal Isabelle has not been loaded. Make sure you start Isabelle with the “-L HOL-Nominal” option.

isabelle emacs -L HOL-Nominal file.thy &

A.2 Can one avoid typing \<guillemotleft>?

If your X-Symbols are correctly installed and enabled in the menu “Proof- General → Options”, then you can just type “<<” (that is two consecutive less-thans). X-Symbols will automatically expand this to \<guillemotleft>. Similarly with \<guillemotright>.

A.3 I want to prove something about support or freshness, but theorems which should be trivial cannot be proved.

Attempting to prove the following lemma lemma support trivial fails:

shows ”supp (App t_1 t_2) = (supp t_1) ∪ (supp t_2)”

will fail. The problem is that supp is a polymorphic constant and this equation is only provable with respect to a concrete atom type. The solution is to give explicit type-annotations, for example

lemma support trivial fails:

shows ”supp (App t_1 t_2) = (supp t_1) ∪ ((supp t_2)::var set)” by (simp add: lam.supp)

Similar type-annotations are necessary in lemmas involving freshness.

Appendix B

Infrastructure

B.1 Equivariance Lemmas

By tagging theorems with the [eqvt]-attribute they are included in the collection of theorems named eqvts. The following general theorems have been tagged by default:

• numbers 0, 1, 2. . . of types int and nat are equivariant π^• number of n = number of n π^• 0 = 0 π^• 1 = 1

• logical connectives are equivariant

π^• (A ∧ B) = (π^• A ∧ π^• B) π^• (A ∨ B) = (π^• A ∨ π^• B) π^• (A −→ B) = (π^• A −→ π^• B) π^• (¬ A) = (¬ π^• A) § π^• (x = y) = (π^• x = π^• y) π^• (if b then e_1 else e_2) = (if π^• b then π^• e_1 else π^• e_2) § π^• a ♯ x = π^• a ♯ π^• x π^• True = True π^• False = False

• units, products, lists and options are equivariant π^• () = () π^• (x, y) = (π^• x, π^• y) π^• fst x = fst (π^• x) π^• snd x = snd (π^• x) π^• [] = [] π^• (x # l) = π^• x # π^• l π^• (l_1 @ l_2) = π^• l_1 @ π^• l_2 π^• Some x = Some (π^• x) π^• None = None

• the following operators on sets are equivariant π^• {} = {} π^• (X ∪ Y) = π^• X ∪ π^• Y § π^• insert x X = insert (π^• x) (π^• X) § π^• (X - Y) = π^• X - π^• Y § π^• (x ∈ X) = (π^• x ∈ π^• X)

π^• set l = set (π^• l)

• the following operators on ints and nats are equivariant π^• Suc x = Suc (π^• x) π^• min x y = min (π^• x) (π^• y) π^• max x y = max (π^• x) (π^• y) π^• (x + y) = π^• x + π^• y π^• (x - y) = π^• x - π^• y π^• (x * y) = π^• x * π^• y π^• (x div y) = π^• x div π^• y

Bibliography

Aydemir, B. E., Bohannon, A., Fairbairn, M., Foster, J. N., Pierce, B. C., Sewell, P., Vytiniotis, D., Washburn, G., Weirich, S., and Zdancewic, S. [2005]. Mechanized Metatheory for the Masses: The PoplMark Challenge. In Proc. of the 18th International Conference on Theorem Proving in Higher- Order Logics (TPHOLs), volume 3603 of LNCS, pages 50–65.

Barendregt, H. [1981]. The Lambda Calculus: Its Syntax and Semantics, volume 103 of Studies in Logic and the Foundations of Mathematics. North- Holland.

Crary, K. [2005]. Logical Relations and a Case Study in Equivalence Checking. In Pierce, B. C., editor, Advanced Topics in Types and Programming Languages, pages 139–160. MIT Press.

Girard, J.-Y., Lafont, Y., and Taylor, P. [1989]. Proofs and Types, volume 7 of Cambridge Tracts in Theoretical Computer Science. Cambridge University Press.

Nipkow, T., Paulson, L. C., and Wenzel, M. [2002]. Isabelle HOL: A Proof Assistant for Higher-Order Logic, volume 2283 of LNCS. Springer-Verlag.

Pitts, A. M. [2003]. Nominal Logic, A First Order Theory of Names and Binding. Information and Computation, 186:165–193.

Pitts, A. M. [2006]. Alpha-Structural Recursion and Induction. Journal of the ACM, 53:459–506.

Pollack, R. [1995]. Polishing Up the Tait–Martin-Löf Proof of the Church- Rosser Theorem. In Proc. of De Wintermöte ’95. Department of Computing Science, Chalmers Univ. of Technology, Göteborg, Sweden.

PoplMark Challenge. http://www.cis.upenn.edu/group/proj/plclub/mmm/.

Takahashi, M. [1995]. Parallel Reductions in Lambda-Calculus. Information and Computation, 118(1):120–127.

Urban, C. [2000]. Classical Logic and Computation. PhD thesis, Cambridge University.

Wenzel, M. [2004]. Using Axiomatic Type Classes in Isabelle. Manual in the Isabelle distribution.

Wenzel, M. [2007]. The Isabelle/Isar Implementation.

Index

\< bullet >, 19 \< sharp >, 19 generate fresh, 27

eqvt attribute, 36
fresh fun, 46
nominal datatype, 30
nominal primrec, 37
Aydemir et al. [2005], 48
Barendregt [1981], 8, 35, 48
Crary [2005], 43, 48
Girard et al. [1989], 48
Nipkow et al. [2002], 3
Pitts [2003], 24, 26, 36
Pitts [2006], 26
Pollack [1995], 6, 48
Takahashi [1995], 6, 48
Urban [2000], 48
Wenzel [2004], 29
Wenzel [2007], 28
PoplMark Challenge [], 48

abstraction, 19
atom decl, 20
atoms, 20

equivariance lemmas, 36

freshness, 19, 26
functions, 37

induction, 33

perm simp, 23
permutations, 21

support, 19, 24

type-classes, 27